(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx opt
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           #\# #\> (lambda (s c p) (declare (ignore c p)) `(values ,(read s nil nil t)))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS
;;;
;;; Arithmetic operations with static modulus
;;;

;; FIXME: Currently MOD* and MOD+ doesn't apply MOD when the number of
;; parameters is one.
(defmacro define-mod-operations (divisor)
  `(progn
     (defun mod* (&rest args)
       (reduce (lambda (x y) (mod (* x y) ,divisor)) args))

     (defun mod+ (&rest args)
       (reduce (lambda (x y) (mod (+ x y) ,divisor)) args))

     #+sbcl
     (eval-when (:compile-toplevel :load-toplevel :execute)
       (locally (declare (muffle-conditions warning))
         (sb-c:define-source-transform mod* (&rest args)
           (if (null args)
               1
               (reduce (lambda (x y) `(mod (* ,x ,y) ,',divisor)) args)))
         (sb-c:define-source-transform mod+ (&rest args)
           (if (null args)
               0
               (reduce (lambda (x y) `(mod (+ ,x ,y) ,',divisor)) args)))))

     (define-modify-macro incfmod (delta)
       (lambda (x y) (mod (+ x y) ,divisor)))

     (define-modify-macro decfmod (delta)
       (lambda (x y) (mod (- x y) ,divisor)))

     (define-modify-macro mulfmod (multiplier)
       (lambda (x y) (mod (* x y) ,divisor)))))


(in-package :cl-user)

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(define-mod-operations +mod+)

(defun calc (num)
  (declare (uint62 num))
  (loop for i from 0
        when (zerop num)
        do (return i)
        do (setq num (mod num (logcount num)))))

(defun main ()
  (let* ((n (read))
         (string-xs (read-line))
         (xs (map 'simple-bit-vector #'digit-char-p string-xs))
         (powers+1 (make-array (+ n 1) :element-type 'uint31 :initial-element 0))
         (powers-1 (make-array (+ n 1) :element-type 'uint31 :initial-element 0))
         (cumuls+1 (make-array (+ n 1) :element-type 'uint31 :initial-element 0))
         (cumuls-1 (make-array (+ n 1) :element-type 'uint31 :initial-element 0)))
    (setf (aref powers+1 0) 1
          (aref powers-1 0) 1)
    (let ((c (count 1 xs)))
      (dotimes (i n)
        (setf (aref powers+1 (+ i 1))
              (mod (* 2 (aref powers+1 i)) (+ c 1))))
      (when (>= c 2)
        (dotimes (i n)
          (setf (aref powers-1 (+ i 1))
                (mod (* 2 (aref powers-1 i)) (- c 1)))))
      (let ((xs (reverse xs)))
        (dotimes (i n)
          (setf (aref cumuls+1 (+ i 1))
                (mod (+ (if (= 1 (aref xs i))
                            (aref powers+1 i)
                            0)
                        (aref cumuls+1 i))
                     (+ c 1))))
        (when (>= c 2)
          (dotimes (i n)
            (setf (aref cumuls-1 (+ i 1))
                  (mod (+ (if (= 1 (aref xs i))
                              (aref powers-1 i)
                              0)
                          (aref cumuls-1 i))
                       (- c 1))))))
      ;; (dbg cumuls+1 cumuls-1)
      (let ((total+ (aref cumuls+1 n))
            (total- (aref cumuls-1 n)))
        (dotimes (i n)
          (println
           (if (zerop (aref xs i))
               (let ((init (mod (+ total+ (aref powers+1 (- n i 1)))
                                (+ c 1))))
                 (+ 1 (calc init)))
               (if (= c 1)
                   0
                   (let ((init (mod (- total- (aref powers-1 (- n i 1)))
                                    (- c 1))))
                     (+ 1 (calc init)))))))))))

#-swank (main)

;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    #+os-windows (run-program "powershell.exe" '("-Command" "Get-Clipboard") :output out :search t)
    #+os-unix (run-program "xsel" '("-b" "-o") :output out :search t)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "200000~%")
    (dotimes (i 200000 (terpri out))
      (write (random 2) :stream out))))

#+swank
(defun bench (&optional (out (make-broadcast-stream)))
  (time (run *dat-pathname* out)))

;; To run: (5am:run! :sample)
#+swank
(it.bese.fiveam:test :sample
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "3
011
"
    "2
1
1
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "23
00110111001011011001110
"
    "2
1
2
2
1
2
2
2
2
2
2
2
2
2
2
2
2
2
2
2
2
1
3
")))


Measured memory_bytes: 30716